
Example Multistate Model Code 

 

;; 1. Based on: 500 

;; 2. Description: Final multistate model with dose introduced on all 

transitions 

;; x1. Author: Uppsala 

 

$SIZES      PD=-500 

$PROBLEM    TRAP 2L_NSCLC multistate model 

$INPUT      C DOMAIN=DROP SEQ USUBJID SUBJID ID SITEID STDY XDATE=DROP 

            CTIME=DROP TIME TAPD TAFT WEEKI DAY CYCLE TIMEN TAPDN 

            EPOCHN AMT RATE DV DV2 SIMFL SIMFL2 VISITT FLAG EOTFL EVID 

            MDV CMT BLOQ FLAGIMP FLLATS NFLLATS TSSRC COHORT TTYPE 

            TTYPE2 TRT01AN AGE SEX RACE RACE2 WT EGFR ALB CRP LDH NLR 

            INR PLAT WBC TB1 TB2 SMOKING PDL1 TGFB1 ECOG METAS METASL 

            MSI HISTOL NNTL CMCOR PMBIO PMAB TMB  

$DATA      ../../../../Data_management/Data/ms-spec16-1.csv IGNORE=@ 

            IGNORE=(SIMFL2.EQ.1) IGNORE=(CMT.EQ.7) WIDE 

$SUBROUTINE ADVAN13 TOL=6 

$MODEL      NCOMP=6 

 

 

$PK 

IF(TRT01AN.EQ.10) CDOSE = 1 

IF(TRT01AN.EQ.11) CDOSE = 0 

 

IF(METAS.EQ.1) METASX=0 

IF(METAS.EQ.0) METASX=1 

 

BSV     = ETA(1) ; place holder 

WEEK=TIME/24/7 

 

IF(NEWIND.NE.2) DV1 = 1 

IF(NEWIND.NE.2) XDV = 1 

IF(NEWIND.NE.2) PPDV = 1 

SPDV = PPDV 

 

IF(NEWIND.NE.2) FLDV3 = 0 

 

; Sim_start: add for simulation 

IF(DV.GE.1) DV1 = DV 

IF(DV.GE.1) XDV = DV 

;Sim_end 

 

;-------------------- GompertzMakeham -------------- 

ALPH = 1.8645e-04 

BETA =0.11425 

CVALUE =0.0007 

GAMA =0.1 

ASMFL=0 

 

X1 = EXP(BETA*AGE-30) 

Y1 = (ALPH*GAMA)/BETA 

GGM = CVALUE + ((ALPH*X1)/(1+(Y1*(X1-1)))) 

 

GMTY = 1/GGM 

 

 



;------------------- Mean transit times ---------------- 

 

MTT12   = THETA(1)*24*7       ;From stable to decrease 

MTT13   = THETA(2)*24*7       ;From stable to progression 

MTT14   = THETA(3)*24*7       ;From stable to unknown 

MTT16   = 52.1429*GMTY*24*7    ;From stable to Death 

 

MTT23   = THETA(4)*24*7       ;From decrease to progression 

MTT24   = THETA(5)*24*7       ;From decrease to unknown 

MTT26   = THETA(6)*24*7    ;From decrease to Death 

 

MTT36   = THETA(7)*24*7       ;From progression to Death 

 

MTT46   = THETA(8)*24*7      ;From unknown to Death 

 

MTT15   = THETA(9)*24*7       ;From stable to dropout 

MTT35   = THETA(9)*24*7       ;From progression to dropout 

MTT45   = THETA(9)*24*7       ;From unknown to dropout 

HZA13   = THETA(10) *EXP(THETA(22)*CDOSE) 

HZA36   = THETA(11) *EXP(THETA(23)*CDOSE)  

 

;--------- Initialize location probabilities ----------- 

IF(NEWIND.NE.2) THEN 

BIO1   = 1 

BIO2   = 0 

BIO3   = 0 

BIO4   = 0 

BIO5   = 0 

BIO6   = 0 

ENDIF 

;------------------------------------------------------- 

;--------- "Bioavailability" factors ------------------- 

 F1    = BIO1 

 F2    = BIO2 

 F3    = BIO3 

 F4    = BIO4 

 F5    = BIO5 

 F6    = BIO6 

;---------------------------------------------------------- 

;--Differential equations for the probability of each score  

HZL13=1/MTT13 

HZL36=1/MTT36 

 

 

 

 

$DES 

DEL = 1E-16 

WB13 =  HZL13*HZA13*(HZL13*(T+DEL))**(HZA13-1) 

WB36 =  HZL36*HZA36*(HZL36*(T+DEL))**(HZA36-1) 

 

K12   = 1/(MTT12) *EXP(THETA(13)*CDOSE) 

K13   = WB13*EXP(METASX*THETA(12))  *EXP(THETA(14)*CDOSE) 

K14   = 1/(MTT14) *EXP(THETA(15)*CDOSE) 

K16   = 1/(MTT16) 

 

K23   = 1/(MTT23) *EXP(THETA(16)*CDOSE) 

K24   = 1/(MTT24) *EXP(THETA(17)*CDOSE) 



K26   = 1/(MTT26) *EXP(THETA(18)*CDOSE) 

 

K36   = WB36  *EXP(THETA(19)*CDOSE) 

K46   = 1/(MTT46) *EXP(THETA(20)*CDOSE) 

 

K15   = 1/(MTT15) *EXP(THETA(21)*CDOSE) 

K25   = 0 ;no observed transition from response to dropout 

K35   = 1/(MTT35) *EXP(THETA(21)*CDOSE) 

K45   = 1/(MTT45) *EXP(THETA(21)*CDOSE) 

 

DADT(1) = - A(1) * (K12 + K13 + K14 + K15 + K16)                            

; STABLE 

DADT(2) =   A(1) * K12 - A(2) * K26 - A(2) * K25 - A(2) * K24 - A(2) * 

K23  ; DECREASE 

DADT(3) =   A(1) * K13 + A(2) * K23 - A(3) * K35 - A(3) * K36               

; PROGRESSION 

DADT(4) =   A(1) * K14 + A(2) * K24 - A(4) * K45 - A(4) * K46               

; UNKNOWN 

DADT(5) =   A(1) * K15 + A(2) * K25 + A(3) * K35 + A(4) * K45               

; DROPOUT 

DADT(6) =   A(1) * K16 + A(2) * K26 + A(3) * K36 + A(4) * K46               

; DEATH 

 

;------------------------------------------------------- 

 

 

 

 

$ERROR 

P_1 = A(1)  ; Probability of observing STABLE 

P_2 = A(2)  ; Probability of observing DECREASE 

P_3 = A(3)  ; Probability of observing PROGRESSION 

P_4 = A(4)  ; Probability of observing UNKNOWN 

P_5 = A(5)  ; Probability of observing DROPOUT 

P_6 = A(6)  ; Probability of observing DEATH 

;------------------------------------------------------- 

 

; ---------Cumulative probabilities ----- 

 

CUP1 = A(1) ; 1 

CUP2 = A(1) + A(2) ; 1 + 2 

CUP3 = A(1) + A(2) + A(3) ; 1 + 2 + 3 

CUP4 = A(1) + A(2) + A(3) + A(4) ; ; 1 + 2 + 3 + 6 

CUP5 = A(1) + A(2) + A(3) + A(4) + A(5); ; 1 + 2 + 3 + 6 

CUP6 = A(1) + A(2) + A(3) + A(4) + A(5) + A(6) ; ; 1 + 2 + 3 + 6   

 

;---- Model prediction ----------------------------------- 

IF(DV.EQ.1.AND.EVID.EQ.0)  Y = P_1 

IF(DV.EQ.2.AND.EVID.EQ.0)  Y = P_2 

IF(DV.EQ.3.AND.EVID.EQ.0)  Y = P_3 

IF(DV.EQ.4.AND.EVID.EQ.0)  Y = P_4 

IF(DV.EQ.5.AND.EVID.EQ.0)  Y = P_5 

IF(DV.EQ.6.AND.EVID.EQ.0)  Y = P_6 

 

TOT = P_1 + P_2 + P_3 + P_4 + P_5 + P_6   

 

 

 



;------------------------------------------------------------ 

; Sim_start: remove for simulation 

;   IF(ICALL.EQ.4) THEN 

; 

;      IF (EVID.EQ.0) THEN 

;      CALL RANDOM (2,R) 

;      USUR=R 

;      ENDIF 

;  

;      DV1=SPDV 

;      IF(VISITT.EQ.1.AND.USUR.LE.CUP1.AND.EVID.EQ.0) DV1=1 

;      IF(VISITT.EQ.1.AND.USUR.GT.CUP1.AND.USUR.LE.CUP2.AND.EVID.EQ.0) DV1=2 

;      IF(VISITT.EQ.1.AND.USUR.GT.CUP2.AND.USUR.LE.CUP3.AND.EVID.EQ.0) DV1=3 

;      IF(VISITT.EQ.1.AND.USUR.GT.CUP3.AND.USUR.LE.CUP4.AND.EVID.EQ.0) DV1=4 

;      IF(VISITT.EQ.1.AND.USUR.GT.CUP4.AND.USUR.LE.CUP5.AND.EVID.EQ.0) DV1=5 

;      IF(VISITT.EQ.1.AND.USUR.GT.CUP5.AND.EVID.EQ.0) DV1=6 

; 

;     XDV  = DV1 

;     IF(SPDV.EQ.5) XDV=5 

;     IF(EVID.EQ.0) PPDV=XDV 

;   ENDIF 

;Sim_end 

;--------- Update Bioavailability factors ------------------- 

 

BIO1=0 

BIO2=0 

BIO3=0 

BIO4=0 

BIO5=0 

BIO6=0 

 

IF(XDV.EQ.1)  BIO1 = 1 

IF(XDV.EQ.2)  BIO2 = 1 

IF(XDV.EQ.3)  BIO3 = 1 

IF(XDV.EQ.4)  BIO4 = 1 

IF(XDV.EQ.5)  BIO5 = 1 

IF(XDV.EQ.6)  BIO6 = 1 

 

 

$THETA  (0,33.8359) ; 1  MTT12 

 (0,11.0384) ; 2  MTT13 

 (0,136.222) ; 3  MTT14 

 (0,72.0868) ; 4  MTT23 

 (0,258.656) ; 5  MTT24 

 (0,540.396) ; 6 MTT26 

 (0,65.0001) ; 7  MTT36 

 (0,133.42) ; 8  MTT46 

 (0,228.091) ; 9  MTT15 MTT35 MTT45 

 (0,0.582216) ; 10 HZA13 

 (0,0.944962) ; 11 HZA36 

 -1.42388 ; 12 METAS13 

 0.227429 ; 13 DOSE12 

 0.433809 ; 14 DOSE13 

 1.02555 ; 15 DOSE14 

 -0.155306 ; 16 DOSE23 

 -0.516235 ; 17 DOSE24 

 0.255654 ; 18 DOSE26 

 0.362582 ; 19 DOSE36 

 0.018097 ; 20 DOSE46 



 -1.12903 ; 21 DOSE5 

 -0.671522 ; 22 DOSE13s 

 -0.568961 ; 23 DOSE36s 

$OMEGA  0  FIX  ; eta as a placeholder 

 

;Sim_start : add/remove for simulation 

;$SIGMA 0 FIX 

;$SIMULATION (12345) (123 UNIFORM) (123 UNIFORM) NSUB=1 ONLYSIM 

$ESTIMATION METHOD=1 LIKE LAPLACE NUMERICAL SLOW PRINT=1 SIGL=9 

            MAXEVALS=9999 CTYPE=4 NSIG=2 MSFO=run509.msf 

$COVARIANCE MATRIX=R 

;Sim_end 

 

$TABLE      ID SUBJID DAY XDV TIME TAFT TRT01AN FLLATS AMT Y EVID MDV 

            CMT DV BIO1 BIO2 BIO3 BIO4 BIO5 BIO6 F1 F2 F3 F4 F5 F6 P_1 

            P_2 P_3 P_4 P_5 P_6 K12 K13 K14 K15 K16 K23 K24 K25 K26 

            K35 K36 K45 K46 NOPRINT FILE=sdtab509 

;vpc run509.mod -samples=500 -dv=XDV -idv=WEEK -rplots=2 -levels=1,2,3,4,5  

;               -flip_com -dir=vpc509vpc -seed=12345 -stratify_on=CDOSE 

 


